Changes and Enhancements to PROC TABULATE

Alan Eaton, SAS Institute Inc.

Abstract

There are several enhancements to PROC TABULATE available in the next major release of the SAS System. This tutorial begins with a review of the fundamental language features of TABULATE. The rest of the tutorial describes enhancements including

- correct interleaving of crossed subgroups
- application of user-defined formats in table cells
- a CONDENSE option to put multiple logical pages on a single physical page
- a BOX option to put page dimension text, a variable label, or a character string in the empty box over the row titles
- using analysis variables in denominator definitions.

The new algorithm for interleaving crossed subgroups is most useful for locating subtotals of classification groups immediately following the levels of each group. Using analysis variables in denominator definitions allows better handling of multiple response variables. Improved algorithms for handling percent calculations now allow the presentation of row, column, and data set percentages in the same table. This tutorial includes detailed examples of how to use many old and new features of PROC TABULATE.

Review of Fundamentals

PROC TABULATE builds tables of descriptive statistics organized into classification hierarchies. Figure 1 displays monthly expenditures and a quarterly summary for departments and accounts of Fleet Footwear, Inc. Totals appear at the top of the table. Sub-totals appear for each department. Table cells are formatted using DOLLAR and COMMA formats.

The principal SAS statements used with PROC TABULATE are

CLASS	supplies classification levels.
VAR	supplies numeric values on which to compute statistics.
TABLE	describes how to classify the data, what values to compute, and how to arrange the values in the table.

Other important statements include KEYLABEL, LABEL, FORMAT, TITLE, FREQ, WEIGHT, and BY.

Contents of a table cell Four things define a table cell:

- classification levels: DEPT ACCT
- a single analysis variable: AMOUNT
- a single statistic: SUM
- a format specification: F=DOLLAR12.2

You can request the same statistics from PROC TABULATE that you can request from PROC SUMMARY. Request statistics by crossing statistic keywords in the TABLE statement.

Table dimensions The first set of examples uses a data set that contains the variables

LAB	a laboratory identification code. Values may be 1, 2, 3, or 4 and are formatted into laboratory location names.
MACHINE	a machine identification code. Each laboratory has two machines, so values may be 1 or 2.
READING	a value read from a machine, for example, blood chemistry data. These values may be continuous.
MONTHNUM	the month during which the reading was made. Values are integers from 1 through 12 and are formatted into month names.

You can create tables of up to three dimensions: column, row, and page. Figure 2 is a one dimensional table. READING is an analysis variable from the VAR statement. LAB is a class variable from the CLASS statement. MEAN is a statistic keyword. All table elements are in the column dimension.

Figure 3 is a two dimensional table using the same elements as before but arranged differently. READING and LAB are in the row dimension; MEAN is in the column dimension. A comma separates dimensions in the TABLE statement. These tables show average machine readings at the four laboratories.

There is a three dimensional table in a later example.

Operations and grouping There are two TABLE statement operations: crossing and concatenation. The crossing operator, *, arranges all levels of one class variable or group within each level of another class variable or under an analysis variable or statistic.

```
READING * LAB
```

Another example is READING * MONTHNUM in Figure 4.

Concatenation means that one group of table cells follows another group of table cells. STD follows MEAN in the column dimension of Figure 4. In the row dimension, READING * MONTHNUM follows READING * LAB.

You can bind the crossing operator to an entire concatenation by grouping the concatenation in parentheses. Note the difference between Figures 4 and 5. Figure 5 has both class variables, LAB and MONTHNUM, grouped under a single instance of READING.

The universal classifier ALL Figure 6 displays MEAN and STD for each machine in each lab. We would like to summarize data for both machines at each lab and for all machines together. Use ALL concatenated with MACHINE to collapse all levels of MACHINE into one classification level under LAB. Concatenate ALL with the entire crossing LAB * (MACHINE ALL) to collapse all distinct levels into a single MEAN and STD as is Figure 7.

Presentation You may want to improve the appearance of this table for a published report. See Figure 8.

- Use PROC FORMAT and FORMAT statements to format class variable values.
- Use LABEL statements to associate text with variable names.
- Use KEYLABEL statements to associate text with statistic keywords and ALL.
- Attach literal modifiers up to forty characters long to any name or keyword in a TABLE statement.
- Use SAS(c) or user-defined formats to print results.

Percentages For the remaining examples we will use another data set that contains data from a hypothetical food product survey. The variables include

ID	survey respondent identification number
AGE	age of respondent
SEX	sex of respondent
OCCUPN	coded value representing respondent's occupation
CEREAL	coded value representing a cereal purchased by the respondent
REASON1- REASON4	coded responses to the question: Why did you buy this cereal? Suggested reasons appear on the survey questionnaire.

You can request two types of percents. PCTN is percentage of N or frequency. PCTSUM is percentage of the SUM of an analysis variable. In each case TABULATE calculates the percentage using the N or SUM value from a particular interaction of class levels, for example, Female 18 to 24 in Figure 9, and a denominator computed by collapsing class levels into a total N or SUM. In Figure 9 collapse levels of SEX * AGE into denominator totals 40, 36, 82, and 158.

You indicate what class levels to collapse by specifying a denominator definition in brackets following PCTN or PCTSUM. Denominator definitions instruct TABULATE how to create ghost crossings to use in collecting denominator totals. Class variables that appear in both the denominator definition and the normal crossing are changed to all in the ghost crossing. Other class variables remain unchanged.

```
TABLE ALL  SEX * AGE,
      (CEREAL ALL) * (N PCTN<SEX*AGE ALL>);

Percent crossings:

   ALL * CEREAL * PCTN <ALL>
      Ghost --> ALL * CEREAL * N
```

```
ALL * ALL * PCTN <ALL>
   Ghost --> ALL * ALL * N
         or  ALL * N

SEX * AGE * CEREAL * PCTN <SEX*AGE>
   Ghost --> ALL * ALL * CEREAL * N
         or  ALL * CEREAL * N

SEX * AGE * ALL * PCTN <SEX*AGE>
   Ghost --> ALL * ALL * ALL * N
         or  ALL * N
```

PCTN<SEX*AGE> collapses all levels of SEX * AGE into a single total for each cereal. This example calculates percentages of column totals by including the entire row dimension in the denominator definition.

A second example, Figure 10, calculates row percents by including the entire column crossing, CEREAL, in the denominator definition. PCTN<CEREAL> collapses all levels of CEREAL for Female 18 to 24 into the denominator total 6. Two females aged 18 to 24 bought Breakfast Fantasy. These two respondents represent 33.33% of the Female 18 to 24 response group.

Note that every crossing that exists in the TABLE statement must be represented among the denominator definitions. The crossings generated by the TABLE statement

```
TABLE (ALL SEX*AGE),
      (CEREAL ALL)*(N PCTN<CEREAL ALL>);
```

are

```
ALL * CEREAL * (N PCTN)
ALL * ALL * (N PCTN)
SEX * AGE * CEREAL * (N PCTN)
SEX * AGE * ALL * (N PCTN)
```

CEREAL appears in two of the crossings but not in the other two. Since we want to preserve the response groups SEX * AGE, use ALL to represent the other two crossings.

You can combine these two percent calculations in a single table by including distinct PCTN specifications in the TABLE statement. See Figure 11.

Recall the variables REASON1 - REASON4. Their values are 1 if the respondent chooses the reason, missing otherwise. In Figure 12 REASON1, labeled 'Flavor', and REASON3, labeled 'Nutritional value', are analysis variables. The SUM of REASON1 values for each cereal is the number of respondents who say that REASON1 motivates their purchase. Specify denominators as before but use PCTSUM to calculate percentage of a total SUM value.

Since no analysis variable appears in the denominator definition, TABULATE uses each analysis variable in turn to choose SUM values. For the crossing CEREAL * SEX * REASON1, the denominator definition is SEX * REASON1, and the ghost denominator crossing is CEREAL * SUM * ALL * REASON1. The ghost crossing is also in the table. The denominator value for Breakfast Fantasy is 27, the sum of REASON1 values for female and male categories. 7 females bought Breakfast Fantasy for its flavor and 7 is 25.9% of all persons who bought Breakfast Fantasy (27). For the crossing CEREAL * SEX * REASON3, the denominator definition is SEX * REASON3, and the denominator is the sum of REASON3 values across both sexes. One of the enhancements is to allow analysis variables to be part of denominator definitions.

Version 5 Enhancements

Interleaving crossed subgroups The current version of TABULATE concatenates whole crossings that crossed subgroups generate. The expression A * (B C) generates the crossings A * B and A * C and concatenates them. The new version interleaves first all levels of B then all levels of C within each level of A. See Figures 13 and 14 for simple examples of the old and new versions, respectively.

The interleaving method is especially useful for displaying subtotals immediately following the values that they summarize. See Figure 1 for examples of subtotals.

Formatted table cells You can use SAS system formats such as DOLLAR and COMMA as well as formats of your own design to format the values in table cells. Table 1 shows examples of SAS system formats. The percent examples show the user-defined format, PCTFMT., at work.

BOX= parameter in TABLE statement BOX specifies the text to be placed in the empty box above the row titles. BOX=_PAGE_ causes the page dimension text to appear in the box. If page dimension text does not fit, it is placed in its normal position, and the box is left empty. BOX=variablename causes the name or label of a variable to appear in the box. BOX='string' causes the quoted string to appear in the box. Any name, label, or quoted string that does not fit is truncated.

CONDENSE option in TABLE statement CONDENSE requests that multiple logical pages be printed on a single physical page. TABULATE continues to condense the output as long as one or more complete logical pages fit on a single printed page. As soon as one logical page exceeds the page size, condensing stops.

Analysis variables in denominator definitions If there are no analysis variables in the denominator definition but there are analysis variables in the TABLE statement, each analysis variable is used in turn with the ghost crossing to choose the N or SUM values. Recall Figure 12. In the current version of TABULATE you can only use each analysis variable in turn.

In the new version you can include analysis variables in denominator definitions to request that the total N or SUM of an analysis variable not in the current crossing be used as the denominator. In the following example make REASON1 the denominator definition, PCTSUM<REASON1>, and percentages are of the total SUM of REASON1 within each response group. See Figure 20.

```
TABLE (ALL CEREAL)*(SUM PCTSUM<REASON1>),
      (ALL SEX) * (REASON1 REASON3);
```

```
ALL * PCTSUM<REASON1> * ALL * REASON1
   Ghost --> ALL * SUM * ALL * REASON1

ALL * PCTSUM<REASON1> * ALL * REASON3
   Ghost --> ALL * SUM * ALL * REASON1
ALL * PCTSUM<REASON1> * SEX * REASON1
   Ghost --> ALL * SUM * SEX * REASON1

ALL * PCTSUM<REASON1> * SEX * REASON3
   Ghost --> ALL * SUM * SEX * REASON1

CEREAL * PCTSUM<REASON1> * ALL * REASON1
   Ghost --> CEREAL * SUM * ALL * REASON1

CEREAL * PCTSUM<REASON1> * ALL * REASON3
   Ghost --> CEREAL * SUM * ALL * REASON1

CEREAL * PCTSUM<REASON1> * SEX * REASON1
   Ghost --> CEREAL * SUM * SEX * REASON1

CEREAL * PCTSUM<REASON1> * SEX * REASON3
   Ghost --> CEREAL * SUM * SEX * REASON1
```

This example gives a measure of the relative importance of flavor and nutrition within each response group. The flavor value is always 100% since the denominator is REASON1 * SUM for the group. Nutrition values are expressed as a percentage of flavor values.

Multiple response variables The following examples show how using analysis variables in denominator definitions opens up new possibilities for handling multiple response variables. Multiple response variables occur frequently on surveys in which one question may have multiple answers.

```
Why did you choose this cereal?
(Circle all reasons that apply.)

   1. Flavor
   2. Texture
   3. Nutritional value
   4. Price
```

The respondent may give one, two, three, or four answers.

Two common strategies exist for handling multiple response variables. The first is to represent each possible response to a question by its own variable on the data set. Each of these variables can have one of two values:

```
1 -- yes, this response has been chosen, or
. -- no, this response has not been chosen.

Sample question: (same as above)

Why did you choose this cereal?
(Circle all reasons that apply.)
 4
   1. Flavor
   2. Texture
   3. Nutritional value
   4. Price
```

When there are many possible responses and only the most important responses are of interest, use the second strategy. Record a respondent's top choices in a small set of variables. Each of these variables can have any of several response values:

```
Sample question:
Indicate from the list below your three most important
reasons for choosing this cereal.

   ____Reason 1     ____Reason 2     ____ Reason 3

Reasons:

   1. Flavor
   2. Texture
   3. Nutritional value
   4. Price
   5. Packaging
   6. Position on shelf
```

For both strategies you want to know the percentage of each response with respect to the number of respondents to the survey. Our example survey studies the reasons that people buy cereal. In particular, the survey examines response groups identified by sex and occupation. The variables SEX and OCCUPN contain this information on the data set.

The first example, Figures 21 and 22, uses strategy one. REASON1, REASON2, REASON3, and REASON4 are restricted to have values 1 or . (missing). ID is a unique number given to each respondent. ID is an analysis variable that is nonmissing on each observation. There is one observation for each respondent, so the crossing ID * N gives a count of all respondents. ID * N is the crossing used to collect denominator values when the denominator definition is SEX * OCCUPN * ID or ALL * ID.

In this example the total number of respondents is 158. 6.3% of respondents are male tinkers who say that at least one of the reasons that they buy a cereal is its flavor. Note that the sum of response percentages for all groups is much greater than 100%. This is the case because each respondent could supply multiple reasons.

Strategy two has respondents choosing the top three of many possible reasons why they buy a cereal. REASON1, REASON2, and REASON3 can have many different values. If there are five possible choices, then on a single observation these variables can have the values

REASON1	REASON2	REASON3
5	2	.
4	1	3
1	1	1
.	.	.

(and so on)

The goal of the analysis is the same: to calculate the percentage of each response with respect to the total number of respondents.

You cannot use REASON1, REASON2, and REASON3 as analysis variables because they do not always represent the same response. You need to translate the values of these three variables into values of a single class variable for TABULATE. For example, an observation that has values 2, 3, 3 for REASON1, REASON2, and REASON3 generates three observations on the new data set. The new variable RESPONSE has value 2 on observation one, 3 on observation two, and 3 again on observation three. The DATA step below accomplishes the transformation:

```
DATA B;
   INPUT ID SEX AGE OCCUPN CEREAL REASON1_REASON3;

   DROP REASON1-REASON3 NORESP FIRST I;
   ARRAY REASONS(3) REASON1-REASON3;

   *---GENERATE A NEW OBSERVATION FOR EACH---;
   *---NON_MISSING REASON VALUE. RESPONSE ---;
   *---ON THE NEW OBSERVATION CARRIES THE ---;
   *---REASON VALUE FROM THE ORIGINAL     ---;
   *---OBSERVATION.                       ---;
   FIRST = 1;
   NORESP= 1;
   DO I = 1 TO 3;
      IF REASONS{I}¬=. THEN DO;
         NORESP = 0;
         RESPONSE = REASONS{I};
         OUTPUT;

         *---ID IS NON-MISSING ONLY ON ---;
         *---FIRST OF NEW OBSERVATIONS ---;
         *---GENERATED FROM CURRENT OLD---;
         *---OBSERVATION.              ---;
         IF FIRST = 1 THEN DO;
            FIRST = 0;  ID = .;
            END;
         END;

   *---GENERATE ONE OBSERVATION IF---;
   *---NO REASON WAS GIVEN.       ---;
   IF NORESP = 1 THEN DO;
      RESPONSE = 999;
      OUTPUT;
      END;
```

Now TABULATE can calculate N for each level of RESPONSE. In order to calculate PCTN on the number of respondents, you need a numeric variable that is nonmissing on the first observation generated from the original three REASON variables but missing on all other new observations generated from the same original observation. N for analysis variables is the frequency of nonmissing values. Therefore, N of ID in the new data set is the number of respondents to the survey. Each original observation represented one respondent.

Now specify denominator definitions that collapse all classification levels so that the denominator value is the count of non-missing ID values on the new data set.

```
PCTN <ALL * RESPONSE * ID>
PCTN <SEX * OCCUPN * RESPONSE * ID>
```

The denominator value is 158 as before. See Figures 23 and 24.

DEPARTMENT AND ACCOUNT NUMBER		FIRST QUARTER				YEAR TO DATE TOTAL
		MONTHLY EXPENDITURES			QUARTERLY TOTAL	
		JANUARY	FEBRUARY	MARCH		
TOTAL		$261,837	$274,331	$248,477	$784,645	$784,645
ACCOUNT-ING	1345	12,980	14,009	17,800	44,789	44,789
	1578	8,000	7,900	4,500	20,400	20,400
	1674	11,950	13,534	17,994	43,478	43,478
	SUB-TOTAL	32,930	35,443	40,294	108,667	108,667
HUMAN RESOURC-ES	2134	34,520	26,560	24,399	85,479	85,479
	2403	10,435	15,494	10,009	35,938	35,938
	SUB-TOTAL	44,955	42,054	34,408	121,417	121,417
SYSTEMS	4138	24,850	22,530	24,399	71,779	71,779
	4279	9,984	14,209	13,500	37,693	37,693
	4290	10,948	14,539	11,459	36,946	36,946
	SUB-TOTAL	45,782	51,278	49,358	146,418	146,418
PRODUC-TION	5139	12,000	14,532	12,098	38,630	38,630
	5260	15,893	14,099	7,304	37,296	37,296
	5370	11,980	13,900	11,480	37,360	37,360
	5399	40,870	38,190	36,106	115,166	115,166
	SUB-TOTAL	80,743	80,721	66,988	228,452	228,452
MARKET-ING	6120	23,435	23,543	19,054	66,032	66,032
	6342	13,049	15,349	18,943	47,341	47,341
	6401	20,943	25,943	19,432	66,318	66,318
	SUB-TOTAL	57,427	64,835	57,429	179,691	179,691

```
TABLE ALL='TOTAL'*F=DOLLAR9.
      DEPT=' '*(ACCT=' ' ALL='SUB-TOTAL')*F=COMMA9.,

      (QTR=' '*(MON='MONTHLY EXPENDITURES'
              ALL='QUARTERLY TOTAL')
       ALL='YEAR TO DATE TOTAL')*BUDGET=' '*SUM=' '

      / BOX='DEPARTMENT AND ACCOUNT NUMBER';
```

FIGURE 1

READING			
LAB			
SAN FRANCISCO	CHICAGO	NEW YORK	HOUSTON
MEAN	MEAN	MEAN	MEAN
0.48	0.51	0.51	0.51

```
TABLE READING*LAB*MEAN;
```

FIGURE 2

		MEAN
READING	LAB	
	SAN FRANCISCO	0.48
	CHICAGO	0.51
	NEW YORK	0.51
	HOUSTON	0.51

```
TABLE READING*LAB, MEAN / RTS=23;
```

FIGURE 3

		MEAN	STD
READING	LAB		
	SAN FRANCISCO	0.48	0.48
	CHICAGO	0.51	0.50
	NEW YORK	0.51	0.50
	HOUSTON	0.51	0.50
READING	MONTHNUM		
	JANUARY	0.50	0.52
	FEBRUARY	0.50	0.49
	MARCH	0.51	0.52
	APRIL	0.51	0.47
	MAY	0.50	0.47
	JUNE	0.52	0.51
	JULY	0.49	0.50
	AUGUST	0.49	0.49
	SEPTEMBER	0.49	0.47
	OCTOBER	0.52	0.52
	NOVEMBER	0.52	0.50
	DECEMBER	0.50	0.46

TABLE READING*LAB READING*MONTHNUM,MEAN STD / RTS=23;

FIGURE 4

		MEAN	STD
READING	LAB		
	SAN FRANCISCO	0.48	0.48
	CHICAGO	0.51	0.50
	NEW YORK	0.51	0.50
	HOUSTON	0.51	0.50
	MONTHNUM		
	JANUARY	0.50	0.52
	FEBRUARY	0.50	0.49
	MARCH	0.51	0.52
	APRIL	0.51	0.47
	MAY	0.50	0.47
	JUNE	0.52	0.51
	JULY	0.49	0.50
	AUGUST	0.49	0.49
	SEPTEMBER	0.49	0.47
	OCTOBER	0.52	0.52
	NOVEMBER	0.52	0.50
	DECEMBER	0.50	0.46

TABLE READING*(LAB MONTHNUM), MEAN STD / RTS=23;

FIGURE 5

LAB	MACHINE	READING MEAN	READING STD
SAN FRANCISCO	1	0.45	0.49
	2	0.52	0.48
CHICAGO	MACHINE		
	1	0.49	0.47
	2	0.53	0.52
NEW YORK	MACHINE		
	1	0.49	0.49
	2	0.54	0.51
HOUSTON	MACHINE		
	1	0.50	0.48
	2	0.52	0.51

TABLE LAB*MACHINE,READING*(MEAN STD) / RTS=23;

FIGURE 6

LAB	MACHINE	READING MEAN	READING STD
SAN FRANCISCO	1	0.45	0.49
	2	0.52	0.48
	ALL	0.48	0.48
CHICAGO	MACHINE		
	1	0.49	0.47
	2	0.53	0.52
	ALL	0.51	0.50
NEW YORK	MACHINE		
	1	0.49	0.49
	2	0.54	0.51
	ALL	0.51	0.50
HOUSTON	MACHINE		
	1	0.50	0.48
	2	0.52	0.51
	ALL	0.51	0.50
ALL		0.50	0.49

TABLE LAB*(MACHINE ALL) ALL,READING*(MEAN STD) / RTS=23;

FIGURE 7

		AVERAGE READING	STANDARD DEVIATION
LABORATORY			
SAN FRANCISCO	MACHINE 1	0.45	0.49
	MACHINE 2	0.52	0.48
	BOTH	0.48	0.48
CHICAGO	MACHINE 1	0.49	0.47
	MACHINE 2	0.53	0.52
	BOTH	0.51	0.50
NEW YORK	MACHINE 1	0.49	0.49
	MACHINE 2	0.54	0.51
	BOTH	0.51	0.50
HOUSTON	MACHINE 1	0.50	0.48
	MACHINE 2	0.52	0.51
	BOTH	0.51	0.50
ALL MACHINES		0.50	0.49

```
TABLE LAB*(MACHINE=' ' ALL='BOTH') ALL='ALL MACHINES',
      READING=' '*(MEAN STD) / RTS=23;

KEYLABEL MEAN='AVERAGE READING'
         STD='STANDARD DEVIATION';
LABEL LAB='LABORATORY';
```

FIGURE 8

		CEREAL							
		BREAKFAST FANTASY		CRUNCHY NUGGETS		FULL-O-FIBER		ALL	
		N	% OF SEX * AGE	N	% OF SEX * AGE	N	% OF SEX * AGE	N	% OF SEX * AGE
ALL		40	100.0%	36	100.0%	82	100.0%	158	100.0%
FEMALE	AGE								
	18 TO 24	2	5.0%	1	2.7%	3	3.6%	6	3.7%
	25 TO 35	.	.	3	8.3%	8	9.7%	11	6.9%
	35 TO 60	6	15.0%	13	36.1%	27	32.9%	46	29.1%
	60 TO 80	3	7.5%	3	8.3%	7	8.5%	13	8.2%
MALE	AGE								
	18 TO 24	2	5.0%	2	5.5%	2	2.4%	6	3.7%
	25 TO 35	2	5.0%	.	.	6	7.3%	8	5.0%
	35 TO 60	18	45.0%	10	27.7%	18	21.9%	46	29.1%
	60 TO 80	7	17.5%	4	11.1%	11	13.4%	22	13.9%

```
TABLE ALL SEX=' '*AGE,
      (CEREAL ALL)*(N*F=4.
                    PCTN<SEX*AGE ALL>='% OF SEX * AGE')
      / RTS=19;
```

FIGURE 9

		CEREAL						ALL	
		BREAKFAST FANTASY		CRUNCHY NUGGETS		FULL-O-FIBER			
		N	% OF CEREAL	N	% OF CEREAL	N	% OF CEREAL	N	% OF CEREAL
ALL		40	25.3%	36	22.7%	82	51.8%	158	100.0%
FEMALE	AGE								
	18 TO 24	2	33.3%	1	16.6%	3	50.0%	6	100.0%
	25 TO 35	.	.	3	27.2%	8	72.7%	11	100.0%
	35 TO 60	6	13.0%	13	28.2%	27	58.6%	46	100.0%
	60 TO 80	3	23.0%	3	23.0%	7	53.8%	13	100.0%
MALE	AGE								
	18 TO 24	2	33.3%	2	33.3%	2	33.3%	6	100.0%
	25 TO 35	2	25.0%	.	.	6	75.0%	8	100.0%
	35 TO 60	18	39.1%	10	21.7%	18	39.1%	46	100.0%
	60 TO 80	7	31.8%	4	18.1%	11	50.0%	22	100.0%

```
TABLE ALL SEX=' '*AGE,
      (CEREAL ALL)*(N*F=4.
                  PCTN<CEREAL ALL>='% OF CEREAL')
      / RTS=19;
```

FIGURE 10

		CEREAL								
		BREAKFAST FANTASY			CRUNCHY NUGGETS			FULL-O-FIBER		
		N	% OF SEX * AGE	% OF CEREAL	N	% OF SEX * AGE	% OF CEREAL	N	% OF SEX * AGE	% OF CEREAL
ALL		40	100.0%	25.3%	36	100.0%	22.7%	82	100.0%	51.8%
FEMALE	AGE									
	18 TO 24	2	5.0%	33.3%	1	2.7%	16.6%	3	3.6%	50.0%
	25 TO 35	.	.	.	3	8.3%	27.2%	8	9.7%	72.7%
	35 TO 60	6	15.0%	13.0%	13	36.1%	28.2%	27	32.9%	58.6%
	60 TO 80	3	7.5%	23.0%	3	8.3%	23.0%	7	8.5%	53.8%
MALE	AGE									
	18 TO 24	2	5.0%	33.3%	2	5.5%	33.3%	2	2.4%	33.3%
	25 TO 35	2	5.0%	25.0%	.	.	.	6	7.3%	75.0%
	35 TO 60	18	45.0%	39.1%	10	27.7%	21.7%	18	21.9%	39.1%
	60 TO 80	7	17.5%	31.8%	4	11.1%	18.1%	11	13.4%	50.0%

```
TABLE ALL SEX=' '*AGE,
      (CEREAL ALL)*(N*F=4.
                  PCTN<SEX*AGE ALL>='% OF SEX * AGE'
                  PCTN<CEREAL ALL>='% OF CEREAL')
      / RTS=19;
```

FIGURE 11

		ALL		
		N	% OF SEX * AGE	% OF CEREAL
ALL		158	100.0%	100.0%
FEMALE	AGE			
	18 TO 24	6	3.7%	100.0%
	25 TO 35	11	6.9%	100.0%
	35 TO 60	46	29.1%	100.0%
	60 TO 80	13	8.2%	100.0%
MALE	AGE			
	18 TO 24	6	3.7%	100.0%
	25 TO 35	8	5.0%	100.0%
	35 TO 60	46	29.1%	100.0%
	60 TO 80	22	13.9%	100.0%

FIGURE 11A

		ALL		SEX			
				FEMALE		MALE	
		FLAVOR	NUTRI-TIONAL VALUE	FLAVOR	NUTRI-TIONAL VALUE	FLAVOR	NUTRI-TIONAL VALUE
ALL	SUM	101	128	44	63	57	65
	% OF THIS CEREAL	100.0%	100.0%	43.5%	49.2%	56.4%	50.7%
BREAKFAST FANTASY	SUM	27	29	7	7	20	22
	% OF THIS CEREAL	100.0%	100.0%	25.9%	24.1%	74.0%	75.8%
CRUNCHY NUGGETS	SUM	22	28	11	16	11	12
	% OF THIS CEREAL	100.0%	100.0%	50.0%	57.1%	50.0%	42.8%
FULL-O-FIBER	SUM	52	71	26	40	26	31
	% OF THIS CEREAL	100.0%	100.0%	50.0%	56.3%	50.0%	43.6%

```
TABLE (ALL CEREAL=' ')*(SUM*F=6.
                        PCTSUM<SEX ALL>='% OF THIS CEREAL'),
      (ALL SEX)*(REASON1 REASON3) / RTS=21;
```

FIGURE 12

		N
A	B	
1	1	1
	2	1
	3	1
2	B	
	1	1
	2	1
	3	1
A	C	
1	1	3
2	C	
	2	3

TABLE A * (B C), N;

FIGURE 13

		N
A	B	
1	1	1
	2	1
	3	1
	C	
	1	3
2	B	
	1	1
	2	1
	3	1
	C	
	2	3

TABLE A * (B C), N;

FIGURE 14

PAGE 1		C	D
B			
1	FREQ	1	1
	SUM	1	1
2	FREQ	1	1
	SUM	1	2
3	FREQ	1	1
	SUM	1	3

```
TABLE A='PAGE ',B='B'*(N='FREQ' SUM),C D
      / RTS=20 BOX=_PAGE_;
```

FIGURE 15

A 1

ROW VARIABLE IS B		C	D
1	FREQ	1	1
	SUM	1	1
2	FREQ	1	1
	SUM	1	2
3	FREQ	1	1
	SUM	1	3

```
TABLE A,B=' '*(N='FREQ' SUM),C D
      / RTS=20 BOX=B;
```

FIGURE 16

A 1

ANALYSIS ON ---->		C	D
1	FREQ	1	1
	SUM	1	1
2	FREQ	1	1
	SUM	1	2
3	FREQ	1	1
	SUM	1	3

```
TABLE A,B=' '*(N='FREQ' SUM),C D
      / RTS=20 BOX='ANALYSIS ON ---->';
```

FIGURE 17

PAGE 1		C	D
B			
1	FREQ	1	1
	SUM	1	1
2	FREQ	1	1
	SUM	1	2
3	FREQ	1	1
	SUM	1	3

PAGE 2		C	D
B			
1	FREQ	1	1
	SUM	2	1
2	FREQ	1	1
	SUM	2	2
3	FREQ	1	1
	SUM	2	3

```
TABLE A='PAGE ',B='B'*(N='FREQ' SUM),C D
      / RTS=20 BOX=_PAGE_ CONDENSE;
```

FIGURE 18

		ALL		SEX			
				FEMALE		MALE	
		FLAVOR	NUTRI-TIONAL VALUE	FLAVOR	NUTRI-TIONAL VALUE	FLAVOR	NUTRI-TIONAL VALUE
ALL	SUM	101	128	44	63	57	65
	% OF FLAVOR	100.0%	126.7%	100.0%	143.1%	100.0%	114.0%
BREAKFAST FANTASY	SUM	27	29	7	7	20	22
	% OF FLAVOR	100.0%	107.4%	100.0%	100.0%	100.0%	110.0%
CRUNCHY NUGGETS	SUM	22	28	11	16	11	12
	% OF FLAVOR	100.0%	127.2%	100.0%	145.4%	100.0%	109.0%
FULL-O-FIBER	SUM	52	71	26	40	26	31
	% OF FLAVOR	100.0%	136.5%	100.0%	153.8%	100.0%	119.2%

```
TABLE (ALL CEREAL=' ')*(SUM*F=6.
                    PCTSUM<REASON1>='% OF FLAVOR'),
      (ALL SEX)*(REASON1 REASON3) / RTS=21;
```

FIGURE 20

CEREAL SURVEY RESPONSE GROUPS: SEX BY OCCUPATION		NUMBER OF RESPON-DENTS	% OF RESPONDENTS BUYING FOR EACH REASON			
			FLAVOR	TEX-TURE	NUTRI-TIONAL VALUE	PRICE
ALL GROUPS		158	63.9%	46.8%	81.0%	43.0%
MALE	TINKER	16	6.3%	5.0%	7.5%	5.6%
	TAILOR	31	14.5%	8.2%	16.4%	8.2%
	SOLDIER	22	9.4%	6.9%	9.4%	9.4%
	STATISTICIAN	13	5.6%	5.0%	7.5%	1.2%
FEMALE	TINKER	17	5.6%	6.9%	8.2%	5.0%
	TAILOR	22	9.4%	5.6%	10.1%	6.9%
	SOLDIER	26	8.2%	6.9%	15.1%	5.0%
	STATISTICIAN	11	4.4%	1.8%	6.3%	1.2%

```
TABLE ALL SEX=' '*OCCUPN=' ',
      ID*N=' '*F=7.
      PCTN<ALL*ID SEX*OCCUPN*ID>*
         (REASON1 REASON2 REASON3 REASON4)
      / RTS=27
BOX='CEREAL SURVEY RESPONSE GROUPS: SEX BY OCCUPATION';
```

FIGURE 21

CEREAL SURVEY RESPONSE GROUPS: SEX BY OCCUPATION (RAW DATA)		NUMBER OF RESPON-DENTS	NUMBER OF RESPONDENTS FOR EACH REASON			
			FLAVOR	TEX-TURE	NUTRI-TIONAL VALUE	PRICE
ALL GROUPS		158	101	74	128	68
MALE	TINKER	16	10	8	12	9
	TAILOR	31	23	13	26	13
	SOLDIER	22	15	11	15	15
	STATISTICIAN	13	9	8	12	2
FEMALE	TINKER	17	9	11	13	8
	TAILOR	22	15	9	16	11
	SOLDIER	26	13	11	24	8
	STATISTICIAN	11	7	3	10	2

```
TABLE ALL SEX=' '*OCCUPN=' ',
      ID*N=' '*F=7.
      N*F=6.*(REASON1 REASON2 REASON3 REASON4)
      / RTS=27
      BOX=
'CEREAL SURVEY RESPONSE GROUPS: SEX BY OCCUPATION (RAW DATA)'
```

FIGURE 22

CEREAL SURVEY RESPONSE GROUPS: SEX BY OCCUPATION		NUMBER OF RESPON-DENTS	% OF RESPONDENTS BUYING FOR EACH REASON			
			FLAVOR	TEX-TURE	NUTRI-TIONAL VALUE	OTHER
ALL GROUPS		158	63.9%	46.8%	81.0%	34.8%
MALE	TINKER	16	6.3%	5.0%	7.5%	5.6%
	TAILOR	31	14.5%	8.2%	16.4%	6.3%
	SOLDIER	22	9.4%	6.9%	9.4%	7.5%
	STATISTICIAN	13	5.6%	5.0%	7.5%	1.2%
FEMALE	TINKER	17	5.6%	6.9%	8.2%	3.7%
	TAILOR	22	9.4%	5.6%	10.1%	5.0%
	SOLDIER	26	8.2%	6.9%	15.1%	3.7%
	STATISTICIAN	11	4.4%	1.8%	6.3%	1.2%

```
TABLE ALL SEX=' '*OCCUPN=' ',
      ID*N=' '*F=7.
      PCTN<ALL*RESPONSE*ID SEX*OCCUPN*RESPONSE*ID>
         *RESPONSE=' '
      / RTS=27
BOX='CEREAL SURVEY RESPONSE GROUPS: SEX BY OCCUPATION';
```

FIGURE 23

CEREAL SURVEY RESPONSE GROUPS: SEX BY OCCUPATION (RAW DATA)		NUMBER OF RESPON-DENTS	NUMBER OF RESPONDENTS FOR EACH REASON			
			FLAVOR	TEX-TURE	NUTRI-TIONAL VALUE	OTHER
ALL GROUPS		158	101	74	128	55
MALE	TINKER	16	10	8	12	9
	TAILOR	31	23	13	26	10
	SOLDIER	22	15	11	15	12
	STATISTICIAN	13	9	8	12	2
FEMALE	TINKER	17	9	11	13	6
	TAILOR	22	15	9	16	8
	SOLDIER	26	13	11	24	6
	STATISTICIAN	11	7	3	10	2

```
TABLE ALL SEX=' '*OCCUPN=' ',
      ID*N=' '*F=7. N*F=6.*RESPONSE=' '
      / RTS=27
      BOX=
'CEREAL SURVEY RESPONSE GROUPS: SEX BY OCCUPATION (RAW DATA)';
```

FIGURE 24